Lab Notes

Blessed are the SEQUE.

By: John S. Simonton, Jr.

Last time, we looked at the live performance features of SEQUE 1.0. Now we turn our attention to the studio-oriented options offered by this "universal" monotonic sequencer program.

Some of the distinctions between stage and studio use are somewhat arbitrary.

For example:

EVENT PROGRAM

The real-time SCORE melody programming mode that we examined in this first section of this piece can obviously be used in a recording studio as well as it can on stage, providing that you're interested in recording only those things that are within the limits of your physical abilities. But the real promise of a small studio (or a big one, for that matter) is that it allows us to produce music that we don't have the chops to do in real time. After all, not everyone has the hours per day that it takes to gain physical mastery of a keyboard - but that doesn't mean that we don't have valid musical ideas, only that we need a little help in expressing them.

If a recording studio is a single thing, it's a time machine that allows days or weeks of work to be compressed into a few minutes of music. One of the programming modes that we have available (EVENT) is specifically designed to operate in this type of time-compression environment. In this mode we enter the music not so much as a melody, but as a series of notes and rests. A series of events which, when reproduced by the computer, turn out to be a melody (maybe).

There is of course nothing new about this mode of operation, this is the way sequencers have always worked. About the only new part is that instead of entering the events as positions of a knob or a series of numbers, we have an AGO keyboard on which to program.

Touching the command keyboard's PROGRAM EVENT pad puts us in this programming mode. (See Figure 1.) Melody lines are entered much as they were with the SCORE mode, except that the computer is no longer watching for how long we hold a key down or how rapidly the notes are played. It is now only interested in whether a key is up or down.



Figure 1

One of the major implications of this is that notes in the melody are "jammed" together in time, and on playback will come out exactly equally spaced, one note per beat. While this is OK in some cases, as a general rule it is unacceptable; because it is unacceptable, we have a REST pad on the command keyboard. The REST pad provides for syncopation. It is a means of "extending" an event so that it takes more than a single beat.

If you're familiar with the operation of the rest key on something like PAIA's Programmable Drum Set, you already have a good idea what's going on, but there still are some surprises here.

Your first thought may be that when you press and release a key on the AGO keyboard, that constitutes an event. Actually, it's two events as far as SEQUE 1.0 is concerned - the first when the key was pressed and the next when it was released. It's important to keep in mind that the REST pad can extend either of these events.

For example, this simple phrase:

Figure 2

would be entered from the keyboard by pressing F and releasing, press A and release, press C, release, press D, release, press F and while holding the F key down, hit the REST block on the keypad, release the F key, tap the REST block, play A, touch the rest block before letting up the A key, release the key, and hit the rest block once more. The measure is now completely entered, and may be played by using the REPEAT or SINGLE keys as described last time. Note particularly that on the fifth note (the second F)

where we wanted to extend the note to a full beat, the REST pad had to be touched twice; once to extend the "key down" event and again to extend the "key up" event.

At first, having to enter two RESTs when we actually want to extend a note for a single beat may seem a pain in the neck (undeniably, it is) but the slight inconvenience buys us a number of things. For example, the ability to slur notes.

In the above example, the D could have been slurred into the F by first touching the REST pad before releasing the D key. This will lengthen the note to occupy the time normally used when the key is released. Then press the F key before releasing the D. This will cause the D to be entered in the next time slot without any articulation (triggering). Now, while holding the F key, touch the REST pad to lengthen it to a quarter note as covered earlier. After releasing the key, enter the additional REST required and proceed as usual.

Having each REST pad activation correspond to a "half" event (kind of) also allows us to produce dotted notes as the exceptions that they are rather than having to make specific tempo provisions for them which must be carried over to all notes in the sequence.

It is also possible to generate articulation changes whenever a note is extended beyond a basic "dual" event. If, for example, you are generating a series of notes where each note uses a key depression plus a REST and a key release plus a REST (four events), these notes can be performed in three different manners. If entered as listed above, the note has equal time allotted for note performance and release. For a staccato style, the note could be entered with a key depression, release, and then two RESTs. For legato styles, the two RESTs could be entered while the key is held down, yielding three "on" events and one "off" event. Each of the above would occupy the same execution time during playback, but would reflect the different articulation styles.

Once the melody is in the computer's memory, it makes no difference whether it got there with SCORE or EVENT programming modes as far as the playback and options are concerned. All of these features (real time or programmed transpositions, single or repeat play, tempo up and down, and tape saves or loads, etc.) work the same.

CLICK TRACK SYNCH

Even more powerful in the studio than the EVENT programming mode are the features added by two other command pads; CLIK and (in the option box) SYNC. These provide a means of synchronizing multiple tracks of sequencer operation.

Once you start using a sequencer for recording, you begin to find more and more places where it can be used to relieve a lot of tedium. The problem in the past has been that it is, for all practical purposes, impossible to manually synchronize a sequencer to a track that's already on tape. Even slight differences in tempo soon build up to an intolerable variation in when a note is supposed to happen and when it actually does happen. Maybe there are people who could manually twiddle a tempo knob and keep things locked together, but that's a hassle.

Most of us are familiar with the classical "click track" approach in which a metronome-like "tick" is recorded on one track of a tape so live musicians can easily maintain the tempo of the original work in over-dubs. Our CLIK and SYNC command pads are simply this old concept extended into the realm of automation.

Touching the CLIK pad causes SEQUE 1.0 to begin producing a very rapid series of "clicks" that are machine readable and represent a standard clock rate which the SYNC option can read and synchronize to. The click appears at the normal cassette output jack (where programs, etc., that are to be saved to tape come from) and when using this option, this output is tied to one of the channels of the tape recorder on which you're recording your audio tracks.

To use the click track option, the tape that you will be recording and mixing your audio onto must always be prepared first; you can't record a lead part and then come back and lay down the click, it won't work like that. Before doing anything else, connect the 8700's cassette output to the input of one track of your recorder, start the tape rolling in record mode, and after allowing a comfortable quiet leader, punch the CLIK pad. Allow the tape to run much longer than you think you'll ever need for what you're going to be recording, one thing you don't want to do is run out of click in the middle of things.

Synchronizing to the click track is simply a matter of connecting the output of the tape channel that contains the click to the normal cassette input jack of the computer, but note that some juggling of the record and playback levels of this channel may be necessary for the computer to properly write and read the channel. In many cases, unless your recorder is capable of providing very high outputs (similar to the earphone levels from the cassette recorders which the computer was designed to work with), you may need to use a small external amp to provide the extra gain and current drive required. If your SYNC fails to respond, try using the earphone jack signal usually provided on multi-track recorders. If this doesn't provide enough power, try using a small portable practice amp (such as a Pygmy or Pignose) whose earphone output should adequately drive the cassette input jack of the computer.

Assuming that you have some rhythm sequence (ordinarily the first laid down) in the computer memory and that you're getting ready to record it as audio, proceed by first punching into the T-SEQ option (if you plan to use it) then touch the SYNC control pad. Roll the tape with the click track channel set to playback and the audio going to one of the other tracks which is naturally in record mode. Before the quiet leader ends, touch the REPT/PLAY command pad and hold it. When the click track starts, so will the sequence. When enough of the track is laid down, terminate the play mode by touching the NORMAL pad.

It is necessary to select the SYNC OPTION last in the above sequence of events because once this option is asserted, a click track must be coming in on the cassette port for the computer to recognize any further commands. If you find yourself with a "dead" computer caused by CLIK being selected with no click track present, you can either run a tape which has a click track or

reset the computer and run the program again.

In situations where the sequence is not to be played from the first down-beat, the SYNC OPTION should be enabled before rolling the tape and REPT/PLAY punched in when the time comes for the sequence to start.

A little constructive play will go a long way toward familiarizing you with the capabilities of this powerful option. Here are some we haven't mentioned yet:

You have probably already noticed the somewhat cryptic METR designations that appear in both the OPTION and TEMPO control boxes. And probably you've figured out that it means metronome (a handy thing in any studio). But this is kind of a super metronome because not only does it have a "pendulum" (which shows in the computer's twin displays) and an audible click (which you hear from the beeper) but it also provides an electrical output in the form of a short positive going pulse that appears as D7 of the D/A output channel (which in turn shows up on the Flag 2 pin jack of the D/A's front panel). This pulse is enormously useful in synchronizing external devices (a Programmable Drum Set, for example).

Since both the SYNC and METR options may be asserted at the same time, the external device can be synched to a pre-recorded audio track.

The METR pad in the TEMPO control box is obviously the tempo control for the metronome. Like the other tempo controls that we looked at last time, this one works in octaves. Each time the pad is touched the metronome tempo doubles until the maximum rate is reached, then the next touch causes the tempo to "fold back" to the minimum rate.

It may be somewhat out of sequence (?) to mention here that the tempo of the metronome is the tempo at which sequences stored in EVENT mode will play back, though of course, the TEMPO UP and DOWN command pads will also alter the tempo of the sequence once saved, as outlined last time.

Another point - When electrically synchronizing things to the click track, the METR TEMPO can still be varied to accomodate different timings, and since it operates by octaves the integrity of the timing will be preserved.

And a hint - the metronome "beep" can also be recorded on tape to provide a "human readable" click track (though it must be saved on a different track than the CLIK).

The only other command pads that we've added are STOP/STEP (a means of stopping the sequence without "forgetting" where we were as well as single stepping through the sequence) and CONT (continue) which allows us to pick up from the point where we STOPped. This feature can provide easy introductions to songs. STOP/STEP through the piece until you reach the REST just prior to the point where the introduction should start. When the CONTINUE pad is touched, the introduction will play, leading into the [illegible] sequence.

I wish I had the space and time (and for that matter, knowledge) to go into some expository statements on the art of small studio multi-tracking, but I leave that to an old friend and new-comer to Polyphony's pages, Craig Anderton. I hope that Craig's and my work will complement one another in this area - I think it will.

I also wish I had the space to go into a detailed analysis of how SEQUE 1.0 works. I don't. If you're really interested, the documented assembler listing which follows is tremendously meaty (though sketchy in parts). Careful study of the code used, in conjunction with the comments given, should be valuable in learning more about software generation and execution.

SEQUE 1.0
COMMAND SUMMARY

PROGRAM

SCORE - Saves melody sequence in real time.

EVENT Saves melody sequence as regularly spaced events.

TRANSPOSE - Saves transpose sequence as events.

PLAY

REPEAT - Plays sequence from beginning, cycles until stopped.

SINGLE - Waits for key on AGO then plays sequence from the beginning. Stops at end of melody.

STOP/STEP - Allows stops or pauses during playback.

CONTINUE - Starts melody playback from where you are in memory.

TAPE

SAVE - Dumps current Melody and Transpose sequences to mag. tape.

LOAD - Loads M & T sequences from tape.

OPTIONS

TABLE - Selects transpose sequence table as source of transpositions (otherwise AGO is source).

METRONOME - Initiates visual metronome display and a "beep".

SYNC. - Shuts down internal timing and accepts pre-recorded click-track for timing information.

CANCEL - Turns all selected options off.

TEMPO

UP - Doubles tempo of melody sequence.

DOWN - Halves tempo of melody sequence.

METRONOME - Doubles speed of metronome display and "beep".

MISC

NORMAL - The "normal synthesizer" mode. Does not alter stored sequences.

```
0010  ;*********************************
0020  ;*                               *
0030  ;*          SEQUE 1.0            *
0040  ;*                               *
0050  ;* MONOTONIC SEQUENCER PROGRAMS *
0060  ;*                               *
0070  ;*              BY               *
0080  ;*    JOHN S. SIMONTON, JR.      *
0090  ;*                               *
0100  ;*(C) 1978 PAIA ELECTRONICS, INC*
0110  ;*     ALL RIGHTS RESERVED       *
0120  ;*                               *
0130  ;*********************************
0140  ;
0150  ;DEFINE ADDRESSES OF LABELS
0160  ;
0170  BEEP  .DL  1F22
0180  DECD  .DL  1F00
0190  CASS  .DL  1EAA
0200  DBIT  .DL  1E49
0210  SBIT  .DL  1E25
0220  OUTP  .DL  0840
0230  DSP   .DL  0820
0240  KBD   .DL  0810
0250  ;
0260  MTB3  .DL  0303
0270  MTB2  .DL  0302
0280  MTB1  .DL  0301
0290  MTBL  .DL  0300
0300  TTBL  .DL  02C0
0310  ;
0320  BUFF  .DL  00F0
0330  KBUF  .DL  00EC
0340  PBUF  .DL  00EB
0350  MPNT  .DL  00EA
0360  TPNT  .DL  00E9
0370  MEND  .DL  00E8
0380  TEND  .DL  00E7
0390  TRNS  .DL  00E6
0400  CNTR  .DL  00E5
0410  TTRN  .DL  00E4
0420  LSTL  .DL  00E3
0430  STUS  .DL  00E2
0440  TPO   .DL  00E1
0450  METF  .DL  00E0
0460  MTRC  .DL  00DF
0470  DUMY  .DL  0003
0480  ;
0490  ;
0500  ;
0510        .OR  1000
0520  ;
1000- A9 00   0530  STAR LDA  00        ;START / RESTART
1002- 85 E2   0540       STA  *STUS     ;CANCEL OPTIONS
1004- A9 0C   0550       LDA  0C        ;NRML COMMAND LINK
1006- 8D 7B 11 0560      STA  ACTN+01   ;PLACE COMMAND LINK
1009- 4C 18 11 0570      JMP  COM       ;JUMP TO COMMON
0580  ;
0590  ;NORMAL OPERATING MODE - DOES NOT ALTER
0600  ;T-SEQUENCE OR M-SEQUENCE
0610  ;
100C- B0 05   0620  NRML BCS  NRM1      ;FIRST PASS THROUGH
100E- 85 E6   0630       STA  *TRNS     ;ZERO TRANSPOSE
1010- 8D 20 08 0640      STA  DSP       ;AND DISPLAYS
1013- A5 EC   0650  NRM1 LDA  *KBUF     ;CHECK FOR NOTES
1015- D0 04   0660  NRM2 BNE  STOR      ;ZERO- NO NEW KEY
1017- A5 EB   0670       LDA  *PBUF     ;SO GET OLD KEY
1019- 29 3F   0680       AND  3F        ;CLEAR BOTH FLAGS
101B- 85 EB   0690  STOR STA  *PBUF     ;SAVE AGAIN
101D- 60      0700       RTS            ;AND RETURN
0710  ;
0720  ;PROGRAM TRANSPOSE MODE - NOTE PLAYED
0730  ;IS "KILLED" WHEN KEY IS RELEASED
0740  ;
101E- B0 0A   0750  TLOD BCS  TL1       ;FIRST PASS, INITIALIZE
1020- 85 E6   0760       STA  *TRNS     ;ZERO TRANSPOSE FIGURE
1022- 85 EB   0770       STA  *PBUF     ;ZERO OUTPUT NOTE
1024- 85 E7   0780       STA  *TEND     ;ZERO TABLE END POINTER
1026- A9 80   0790       LDA  80        ;TURN T-SEQUE OPTION
1028- 85 E2   0800       STA  *STUS     ;ON
102A- A6 E7   0810  TL1  LDX  *TEND     ;GET TRANSPOSE POINTER
102C- 8E 20 08 0820      STX  DSP       ;SHOW IT
102F- A5 EC   0830       LDA  *KBUF     ;GET THE NOTE
1031- F0 06   0840       BEQ  TL2       ;ZERO- NO KEY, SAVE
1033- C5 EB   0850       CMP  *PBUF     ;KEY SAME AS LAST?
1035- F0 05   0860       BEQ  TRTN      ;YES - LEAVE
1037- E6 E7   0870       INC  *TEND     ;POINT TO NEXT LOCATION
1039- 9D C0 02 0880 TL2  STA  TTBL,X    ;SAVE TRANSPOSE
103C- 85 EB   0890  TRTN STA  *PBUF     ;AND OUTPUT AS NOTE
103E- 60      0900       RTS            ;THEN RETURN
0910  ;
0920  ;PROGRAM SCORE MODE - USES REAL-TIME CLOCK
0930  ;
103F- 20 84 11 0940  MSAV JSR  MSV1     ;CALL SAVE MODULE
1042- E6 E5   0950       INC  *CNTR     ;INCREMENT THE TEMPO
1044- 60      0960       RTS            ;COUNTER AND RETURN
0970  ;
0980  ;CONTINUE PLAY MODE - DOES NOT RESET
0990  ;M-SEQUENCE OR T-SEQUENCE POINTERS
1000  ;
1045- 38      1010  CNTU SEC            ;SKIP INITIALIZATION
1020  ;
1030  ;REPEAT PLAY MODE - WHEN FIRST ENTERED
1040  ;M-SEQ AND T-SEQ POINTERS ARE SET TO ZERO
1050  ;BY THE PLAY MODULE (PLA1)
1060  ;
1046- 20 AC 11 1070  RPLA JSR  PLA1     ;CALL PLAY MODULE
1049- AD 14 11 1080       LDA  STBL+14   ;WAS THE PREVIOUS MODE
104C- C5 E3   1090       CMP  *LSTL     ;MSAV (PROG. SCORE)?
104E- D0 02   1100       BNE  RPL1      ;NO-SKIP INCREMENT
1050- E6 E9   1110       INC  *TPNT     ;INC. T-SEQ POINTER
1052- 24 E2   1120  RPL1 BIT  *STUS     ;T-SEQ ASSERTED ?
1054- 30 0A   1130       BMI  ROUT      ;OPTION ON - LEAVE
1056- A5 EC   1140       LDA  *KBUF     ;OPTION OFF- GET NOTE
1058- F0 02   1150       BEQ  OLDK      ;AND IF NO NOTE, BRANCH
105A- 85 E4   1160       STA  *TTRN     ;SAVE NOTE FOR NEXT TIME
105C- A5 E4   1170  OLDK LDA  *TTRN     ;GET LAST ACTIVE NOTE
105E- 85 E6   1180       STA  *TRNS     ;USE AS TRANSPOSE
1060- E6 E5   1190  ROUT INC  *CNTR     ;INCREMENT TEMPO COUNTER
1062- 60      1200       RTS            ;AND RETURN
1210  ;
1220  ;SINGLE PLAY MODE - WAITS FOR AGO KEY
1230  ;THEN PLAYS SEQUENCE ONCE THROUGH
1240  ;TRANSPOSED TO INDICATED KEY
1250  ;
1063- 90 04   1260  SING BCC  SNG1     ;FIRST PASS, BRANCH
1065- A5 EC   1270       LDA  *KBUF     ;AGO KEY DOWN ?
1067- D0 DD   1280       BNE  RPLA      ;YES - PLAY SEQUENCE
1069- 20 46 10 1290  SNG1 JSR  RPLA     ;NO - "PLAY" THEN RETURN
106C- A5 EA   1300       LDA  *MPNT     ;M-SEQ POINTER > 0 ?
106E- D0 0B   1310       BNE  SRTN      ;YES - RETURN
1070- A9 00   1320       LDA  00        ;NO - PREPARE
1072- 85 E5   1330       STA  *CNTR     ;ZERO TEMPO COUNTER
1074- A6 E8   1340       LDX  *MEND     ;POINT TO LAST NOTE
1076- BD 01 03 1350      LDA  MTB1,X    ;OF M-SEQ AND GET IT
1079- 85 EB   1360       STA  *PBUF     ;PLACE IN PLAY BUFFER
107B- 60      1370  SRTN RTS            ;THEN RETURN
1380  ;
1390  ;UP TEMPO AND DOWN TEMPO - COMMON PORTION
1400  ;OF BOTH PROGRAMS ON PAGE 2
1410  ;
107C- A9 7E   1420  UTMP LDA  7E        ;THE OP-CODE FOR ROR
107E- D0 02   1430       BNE  U/D       ;BRANCH ALWAYS
1080- A9 3E   1440  DTMP LDA  3E        ;THE OP-CODE FOR ROL
1082- 4C 00 12 1450  U/D  JMP  TCOM     ;JUMP FOR THE REST
1460  ;
1470  ;REST MODE - EXTENDS NOTES OR UN-NOTES
1480  ;WHEN IN PROGRAM EVENT MODE
1490  ;
1085- 18      1500  REST CLC            ;PREPARE FOR ADDITTION
1086- A5 E5   1510       LDA  *CNTR     ;GET TEMPO COUNTER
1088- 65 E1   1520       ADC  *TPO      ;ADD TEMPO VALUE
108A- 85 E5   1530       STA  *CNTR     ;PUT COUNTER BACK
108C- A5 E3   1540       LDA  *LSTL     ;AND RETURN TO
108E- 8D 7B 11 1550      STA  ACTN+01   ;PREVIOUS OPERATING
1091- 60      1560       RTS            ;MODE
1570  ;
1580  ;STOP/STEP MODE - STOPS PLAY WITHOUT
1590  ;CHANGING POINTERS.  SINGLE STEPS THROUGH
1600  ;SEQUENCE
1610  ;
1092- B0 0E   1620  STEP BCS  STP1     ;NOT FIRST PASS-BRANCH
1094- A9 FF   1630       LDA  0FF       ;SET TEMPO COUNTER AT
1096- 85 E5   1640       STA  *CNTR     ;"TIMED OUT" VALUE
```

```
1098-  20 B6 11  1650        JSR CONT     :CALL PART OF PLAY MODULE
1098-  8E 20 08  1660        STX DSP      :DISPLAY M-SEQ POINTER
109E-  A9 00     1670        LDA 00       :MAKE TRANSPOSE VALUE
10A0-  85 E6     1680        STA *TRNS    :EQUAL TO ZERO
10A2-  60        1690  STP1  RTS          :AND RETURN
                 1700  :
                 1710  :PROGRAM EVENT MODE - SAVES M-SEQUENCE
                 1720  :BUT SUBSTITUTES EVENT CLOCK FOR REAL-TIME
                 1730  :CLOCK
                 1740  :
10A3-  B0 02     1750  ESAV  BCS ES1       :FIRST PASS, INITIALIZE
10A5-  85 E5     1760        STA *CNTR    :TEMPO COUNTER AS ZERO
10A7-  20 84 11  1770  ES1   JSR MSV1     :CALL SAVE MODULE
10AA-  A5 E5     1780        LDA *CNTR    :GET TEMPO COUNTER
10AC-  D0 05     1790        BNE EOUT     :NO ENTRY-RETURN
10AE-  18        1800        CLC          :PREPARE
10AF-  65 E1     1810        ADC *TPO     :ADD TEMPO VALUE
10B1-  85 E5     1820        STA *CNTR    :SAVE AS TEMPO COUNTER
10B3-  60        1830  EOUT  RTS          :THEN RETURN
                 1840  :
                 1850  :OPTION MENU - RETURNS TO PREVIOUS
                 1860  :OPERATING MODE AFTER TURNING ON OR
                 1870  :CANCELLING OPTIONS
                 1880  :
10B4-  85 E9     1890  TBLM  STA *TPNT    :T-SEQ POINTER TO BEG
10B6-  A5 E2     1900        LDA *STUS    :ASSERT T-SEQ OPTION
10B8-  09 80     1910        ORA 80
10BA-  D0 0E     1920        BNE MCOM     :BRANCH ASLWAYS
10BC-  A5 E2     1930  MET   LDA *STUS    :TURN METRONOME ON
10BE-  09 40     1940        ORA 40
10C0-  D0 08     1950        BNE MCOM     :BRANCH ALWAYS
10C2-  A5 E2     1960  SYNC  LDA *STUS    :TURN ON SYNC TO
10C4-  09 01     1970        ORA 01       :CLICK TRACK OPTION
10C6-  D0 02     1980        BNE MCOM     :BRANCH ALWAYS
10C8-  A9 00     1990  CNCL  LDA 00       :PREPARE AND
10CA-  85 E2     2000  MCOM  STA *STUS    :CANCEL ALL OPTIONS
10CC-  4C 0F 12  2010        JMP TCM1     :JUMP FOR THE REST
                 2020  :
                 2030  :CLICK MODE - SENDS CLICK TRACK TO TAPE
                 2040  :AGO KEYBOARD SCAN RATE IS TIMER
                 2050  :
10CF-  18        2060  CLIK  CLC          :PREPARE TO SEND "0"
10D0-  20 25 1E  2070        JSR SBIT     :SEND IT
10D3-  60        2080        RTS          :RETURN FOR KEYBOARD DELAY
                 2090  :
                 2100  :METRONOME TEMPO CHANGE - PROGRAM ON PAGE 2
                 2110  :
10D4-  4C 54 12  2120  TCHG  JMP TCH      :JUMP TO PROGRAM
                 2130  :
                 2140  :DUMP M&T-SEQ TO TAPE - PROGRAM ON PAGE 2
                 2150  :
10D7-  4C 20 12  2160  OTAP  JMP TOUT     :JUMP TO PROGRAM
                 2170  :
                 2180  :LOAD M&T-SEQ FROM TAPE - PROGRAM ON PAGE 2
                 2190  :
10DA-  4C 33 12  2200  ITAP  JMP TIN      :JUMP TO PROGRAM
                 2210  :
                 2220  :
                 2290  :COMMAND LINKS -- LOW BYTE OF ADDRESS OF SUBS
                 2300  :
1100- 85 85 85 85 C2 BC B4 C8
1108- CF D4 80 7C DA D7 0C 0C
1110- 45 92 63 46 3F A3 1E 46
                 2790  :
                 2800        .OR 1118
                 2810  :
                 2820  :COMMON PROGRAM - DOES METRONOME WHEN ON
                 2830  :ADDS PLAY AND TRANSPOSE BUFFERS TO GET
                 2840  :OUTPUT NOTE, PLAYS NOTE, READS COMMAND
                 2850  :KEYBOARD AND JUMPS TO SELECTED MODE
                 2860  :SUBSTITUTES CLICK SYNCH FOR KEYBOARD
                 2870  :TIMING LOOP WHEN SYNC OPTION IS ASSERTED
                 2880  :
1118-  A5 E2     2890  COM   LDA *STUS    :CHECK OPTIONS
111A-  48        2900        PHA          :SAVE A COPY
111B-  0A        2910        ASL          :MERONOME ON ?
111C-  10 22     2920        BPL COM0     :NO - BRANCH
111E-  C6 DF     2930        DEC *MTRC    :DECREMENT METRONOME COUNTER
1120-  10 1E     2940        BPL COM0     :NOT <0 YET, BRANCH
1122-  A6 E1     2950        LDX *TPO     :TIME UP, GET TEMPO VALUE
1124-  CA        2960        DEX          :DECREMENT ONCE
1125-  86 DF     2970        STX *MTRC    :THEN SAVE AS COUNTER

1127-  A9 80     2980        LDA 80       :TO DETERMINE ALTERNATE DISPLAY
1129-  AA        2990        TAX          :CYCLE AND "PENDULUM" LEFT
112A-  18        3000        CLC          :PREPARE FOR ADDITTION
112B-  65 E0     3010        ADC *METF    :ADD FLIP-FLOP VALUE
112D-  85 E0     3020        STA *METF    :SAVE NEW VALUE
112F-  10 0C     3030        BPL MET1     :ALTERNATE? - DISPLAY
1131-  A5 EB     3040        LDA *PBUF    :OTHERWISE, GET OUTPUT
1133-  09 80     3050        ORA 80       :SET D7
1135-  85 EB     3060        STA *PBUF    :SAVE IN PLAY BUFFER
1137-  18        3070        CLC          :PREPARE AND
1138-  20 25 1E  3080        JSR SBIT     :CALL BEEP
113B-  A2 08     3090        LDX 08       :"PENDULUM" RIGHT
113D-  8E 20 08  3100  MET1  STX DSP      :SHOW PENDULUM
1140-  A5 E6     3110  COM0  LDA *TRNS    :IS THERE A TRANSPOSE ?
1142-  F0 03     3120        BEQ COM1     :NO - BRANCH
1144-  18        3130  TRAN  CLC          :YES - PREPARE
1145-  69 A4     3140        ADC 0A4      :CALCULATE TRANSPOSE VALUE
1147-  18        3150  COM1  CLC          :MORE PREPARATION
1148-  65 EB     3160        ADC *PBUF    :CALCULATE NOTE
114A-  8D 40 08  3170  COUT  STA OUTP     :PLAY NOTE
114D-  68        3180        PLA          :GET STUS (OPTION CODES)
114E-  6A        3190        ROR          :SYNC OPTION ON ?
114F-  90 06     3200        BCC KRED     :NO - SKIP
1151-  20 49 1E  3210        JSR DBIT     :WAIT FOR CLIK
1154-  4C 6D 11  3220        JMP CTRL     :SKIP READING AGO
1157-  2C 10 08  3230  KRED  BIT KBD      :WAIT FOR DUMMY SCAN
115A-  10 FB     3240        BPL KRED     :LOOP UNTIL STARTED
115C-  AD 10 08  3250  KR2   LDA KBD      :WAIT FOR SCAN TO START
115F-  30 FB     3260        BMI KR2      :LOOP UNTIL STARTED
1161-  2C 10 08  3270  KR3   BIT KBD      :CHECK FOR KEYS DOWN
1164-  30 05     3280        BMI KRTN     :WHEN SCAN DONE, RETURN
1166-  50 F9     3290        BVC KR3      :CURRENT KEY NOT DOWN, LOOP
1168-  AD 10 08  3300        LDA KBD      :KEY DOWN, GET IT
116B-  85 EC     3310  KRTN  STA *KBUF    :SAVE RESULT
116D-  20 00 1F  3320  CTRL  JSR DECD     :GET COMMAND
1170-  B0 06     3330        BCS DO       :OLD COMMAND - DO IT
1172-  B9 00 11  3340        LDA STBL,Y   :NEW COMMAND - GET LINK
1175-  8D 7B 11  3350        STA ACTN+01  :PLACE LINK
1178-  A9 00     3360  DO    LDA 00       :THIS WILL BE HANDY
117A-  20 03 00  3370  ACTN  JSR DUMY     :CALL OPERATING MODE
117D-  AD 7B 11  3380        LDA ACTN+01  :SAVE CURRENT COMMAND
1180-  85 E3     3390        STA *LSTL    :LINK FOR LATER
1182-  D0 94     3400        BNE COM      :AND LOOP ALWAYS
                 3410  :
                 3420  :SAVE MODULE - TAKES CARE OF ALTERNATELY
                 3430  :STACKING DURATIONS AND NOTES IN M-SEQUENCE
                 3440  :USES WHAT WILL BE "END OF SEQUENCE"
                 3450  :INDICATOR IN PLAY MODES AS POINTER
                 3460  :
1184-  B0 09     3470  MSV1  BCS MS1      :FIRST PASS?
1186-  8D 01 03  3480        STA MTBL+01  :YES-ZERO PROGRAM NOTE
1189-  85 E8     3490        STA *MEND    :ZERO M-SEQ POINTER
118B-  85 E6     3500        STA *TRNS    :ZERO TRANSPOSE
118D-  85 EB     3510        STA *PBUF    :ZERO OUTPUT NOTE
118F-  A5 E5     3520  MS1   LDA *CNTR    :GET TIME SINCE LAST NOTE
1191-  A6 E8     3530        LDX *MEND    :AND M-SEQ END POINTER
1193-  9D 00 03  3540        STA MTBL,X   :SAVE THE TIME
1196-  20 13 10  3550        JSR NRM1     :IN CASE NO KEYS DOWN
1199-  29 7F     3560        AND 7F       :CLEAR D7 IN OUTPUT NOTE
119B-  DD 01 03  3570        CMP MTBL,X   :SAME AS LAST NOTE?
119E-  F0 0B     3580        BEQ OUT      :YES, LEAVE
11A0-  E8        3590        INX          :NO, SAVE BY INCREMENTING
11A1-  E8        3600        INX          :M-SEQ POINTER TWICE
11A2-  86 E8     3610        STX *MEND    :AND SAVING AS END
11A4-  9D 01 03  3620        STA MTBL,X   :THEN SAVE NOTE
11A7-  A9 00     3630        LDA 00       :AND ZERO TIME SINCE
11A9-  85 E5     3640        STA *CNTR    :LAST NOTE
11AB-  60        3650  OUT   RTS          :AND RETURN
                 3660  :
                 3670  :PLAY MODULE - MANAGES M-SEQ AND T-SEQ
                 3680  :POINTERS AS WELL AS TEMPO CLOCK.
                 3690  :DETERMINES WHEN NOTES ARE TO BE PLAYED
                 3700  :
11AC-  B0 08     3710  PLA1  BCS CONT     :FIRST PASS ?
11AE-  85 E4     3720        STA *TTRN    :YES-ZERO TEMP. TRANSPOSE
11B0-  85 E9     3730  LP1   STA *TPNT    :ZERO T-SEQ POINTER
11B2-  85 EA     3740  LP2   STA *MPNT    :AND M-SEQ POINTER
11B4-  85 E5     3750        STA *CNTR    :AND CLOCK (TEMPO CONTER)
11B6-  A5 E5     3760  CONT  LDA *CNTR    :GET CLOCK
11B8-  A4 E9     3770        LDY *TPNT    :GET T-SEQ POINTER
11BA-  A6 EA     3780        LDX *MPNT    :GET M-SEQ POINTER
11BC-  DD 02 03  3790        CMP MTB2,X   :TIME UP?
```

```
11BF-  90 15     3800        BCC PL1     :NO, BRANCH
11C1-  A9 00     3810        LDA 00      :YES, PREP  COUNTER, ETC.
11C3-  85 E5     3820        STA *CNTR   :FOR NEXT ACCUMULATION
11C5-  E8        3830        INX         :INCREMENT M-SEQ POINTER
11C6-  E8        3840        INX         :TWICE
11C7-  86 EA     3850        STX *MPNT   :AND SAVE NEW POINTER
11C9-  E4 E8     3860        CPX *MEND   :END OF M-SEQ?
11CB-  D0 09     3870        BNE PL1     :NO - BRANCH
11CD-  C8        3880        INY         :YES, INC T-SEQ POINTER
11CE-  C4 E7     3890        CPY *TEND   :END OF T-SEQ ?
11D0-  B0 DE     3900        BCS LP1     :YES-START T&M-SEQ AGAIN
11D2-  84 E9     3910        STY *TPNT   :NO-SAVE T-SEQ POINTER
11D4-  D0 DC     3920        BNE LP2     :BRANCH-START M-SEQ AGAIN
11D6-  BD 03 03  3930 PL1    LDA MTB3,X  :GET THE NOTE
11D9-  85 EB     3940        STA *PBUF   :SAVE IN PLAY BUFFER
11DB-  B9 C0 02  3950        LDA TTBL,Y  :GET TRANSPOSE
11DE-  85 E6     3960        STA *TRNS   :TO TRANSPOSE BUFFER
11E0-  60        3970        RTS         :RETURN
                 3980 :
                 3990 :TAPE TRANSFER PARAMETER TABLE
                 4000 :
                 4010 TAPE .HS FF00FF03C002C002
                 4020 :
                 4030       .OR 1200
                 4040 :
                 4050 :COMMON PORTION OF TEMPO UP & DOWN -
                 4060 :ROTATES RIGHT OR LEFT THE DURATIONS
                 4070 :SAVED WITH M-SEQUENCE
                 4080 :
1200-  8D 06 12  4090 TCOM STA PLAC   :PLACE ROR OR ROL OP CODE
1203-  A2 00     4100       LDX 00    :ZERO A COUNTER/POINTER
1205-  18        4110 TLP   CLC       :PREPARE
1206-  7E 02 03  4120 PLAC  ROR MTB2,X :ROTATE SAVED TEMPO
1209-  E8        4130       INX       :INCREMENT POINTER TWICE
120A-  E8        4140       INX       :TO POINT TO NEXT
120B-  E4 E8     4150       CPX *MEND :END OF M-SEQ ?
120D-  D0 F6     4160       BNE TLP   :NO - LOOP FOR MORE
120F-  A5 E3     4170 TCM1  LDA *LSTL  :DONE, GET LINK AND
1211-  8D 7B 11  4180       STA ACTN+01 :SET UP FOR PREVIOUS MODE
1214-  60        4190       RTS       :THEN RETURN
                 4200 :
                 4210 :SET UP PROCEDURE FOR TAPE TRANSFER
                 4220 :
1215-  A2 07     4230 STTP  LDX 07     :TRANSFER 7 BYTES
1217-  BD E1 11  4240 STP   LDA TAPE,X :GET PARAMETER FROM TABLE
121A-  95 F0     4250       STA *BUFF,X :PLACE IN POT-SHOT BUFFER
121C-  CA        4260       DEX       :POINT TO NEXT, MORE ?
121D-  D0 F8     4270       BNE STP   :YES - LOOP
121F-  60        4280       RTS       :NO - RETURN
                 4290 :
                 4300 :DUMP M-SEQ AND T-SEQ TO TAPE
                 4310 :
1220-  20 15 12  4320 TOUT  JSR STTP   :SET UP FOR TRANSFER
1223-  A5 E8     4330       LDA *MEND :SAVE M-SEQ END WITH
1225-  8D 00 03  4340       STA MTBL  :M&T-SEQUENCE
1228-  A5 E7     4350       LDA *TEND :ALSO T-SEQUENCE END
122A-  8D 01 03  4360       STA MTB1
122D-  A9 DD     4370       LDA 0DD   :SET UP FOR DUMP
122F-  20 46 12  4380       JSR DOTP  :AND DO IT
1232-  60        4390       RTS       :THEN RETURN
                 4400 :
                 4410 :LOAD M-SEQ AND T-SEQ FROM TAPE
                 4420 :
1233-  20 15 12  4430 TIN   JSR STTP   :SET UP FOR TRANSFER
1236-  A9 11     4440       LDA 11    :SET UP FOR LOAD
1238-  20 46 12  4450       JSR DOTP  :AND DO IT
123B-  AD 00 03  4460       LDA MTBL  :PLACE M-SEQUENCE END
123E-  85 E8     4470       STA *MEND
1240-  AD 01 03  4480       LDA MTB1  :AND T-SEQUENCE END
1243-  85 E7     4490       STA *TEND
1245-  60        4500       RTS       :THEN RETURN
                 4510 :
                 4520 :PERFORM TAPE TRANSFER
                 4530 :
1246-  20 AA 1E  4540 DOTP  JSR CASS   :CALL POT-SHOT
1249-  AD 0F 11  4550       LDA STBL+0F :SET UP TO RETURN
124C-  8D 7B 11  4560       STA ACTN+01 :IN NORMAL MODE
124F-  18        4570       CLC       :PREPARE
1250-  20 22 1F  4580       JSR BEEP  :SIGNAL DONE
1253-  60        4590       RTS       :AND RETURN
                 4600 :
                 4610 :CHANGE METRONOME TEMPO
                 4620 :
1254-  85 DF     4630 TCH   STA *MTRC  :ZERO METRONOME CLOCK
1256-  66 E1     4640       ROR *TPO  :HALVE TEMPO VALUE
1258-  90 02     4650       BCC TCHR  :IF NOT ZERO, LEAVE
125A-  66 E1     4660       ROR *TPO  :ZERO, MAKE NOT ZERO
125C-  D0 B1     4670 TCHR  BNE TCM1  :GO SET UP PREVIOUS MODE
                 4680 :
                 4690 END   .EN
                 4700
```

NOTE: The following is available from PAIA Electronics, PO Box 14359, Oklahoma City, OK 73114:

Seque 1.0 cassette data tape with complete documentation and plastic computer keyboard overlay.....$6.95 postpaid

PAIA P-4700/J POLYPHONIC...Of Course!

SYNTHESIZER/COMPUTER



But that's just the beginning, because only PAIA Synthesizer/Computers Allow you to use any of a growing number of personality programs. Including:

POLYPHONIC

MUS 1.0 - a 16 voice polyphonic synthesizer with software transient generators.

SEQUENCERS

SEQUE 1.0 - a general purpose monotonic sequencer.

POLY SEQUE - a 4 voice sequencer.

COMPOSERS

PINK TUNES- Composes 4 part harmonies

PINK FREUD - Composes 4 part canons.

SPECIAL EFFECTS

SHAZAM - Multiple keyboard split and chorusing.

AND MANY MORE COMING SOON!

The P-4700/J Synthesizer/Computer package includes the following module complement: two 4710 Balanced Modulator VCAs, 4711 Stereo Mixer, 4712 Reverb, three 4720 Wide Range VCOs, two 4730 Multi-Modal VCFs, 4740 ADSR Envelope Generator, 2720-5 Control Oscillator/Noise Source, 8780 Digital to Analog Converter, 8781 QuASH (Quad Addressable Sample & Hold), and the 8782 Intelligent Keyboard with 8700 Computer Controller housed in sturdy vinyl covered road cases.

System firmware includes: PIEBUG - system monitor; POT-SHOT - cassette interface and MUS 1.0 synthesizer operating system.

P-4700/J Synthesizer/Computer Kit $749.00
Includes all listed software (shipped freight collect)

( ) Sounds intriguing, but I need a lot more information. Please send the most recent edition of your "Friendly Stories About Computers/Synthesizers" .....$3.00 postpaid

( ) Please also send complete instruction manual set for the P-4700/J .....$10.00 (refundable upon purchase of P-4700/J Kit)

( ) I've been with you all along. Please send complete P-4700/J Synthesizer/Computer Kit. .....$749.00 .....shipped freight collect .....

Name: _____

Address: _____

City: _____ State: _____ Zip: _____

Visa/BAC_____ Master Charge_____ Card No. _____

Expiration date: _____ _____

PAIA ELECTRONICS DEPT. 1020 W. WILSHIRE BLVD., OKLAHOMA CITY, OK 73116 (signature)



Typical control panel configuration

ONLY FROM:

PAIA ELECTRONICS, INC.
1020 WEST WILSHIRE BLVD.
OKLAHOMA CITY, OKLAHOMA 73116
(405) 843-9626