

POLY-SPLIT

Many times we've talked about how the personality of our computer based equipment is a function of the operating system software that we happen to be running at the moment. Let's play some head games with the gear and feed it some code that will give it a split personality.

POLY-SPLIT does just that; it gives us two complete polyphonic synthesizer systems under the control of one keyboard. Play a chord or note on the lower keys and they are always assigned to a lower group of outputs. Play on higher keys and the result is assigned to another group.

Before we get into the listing of this program and its operation and use, we need to keep one fact clearly in mind; POLY-SPLIT is simply an extension of the polyphonic personality offered in MUS 1.0. All of the options offered by that code (STGs, dynamic output refresh, etc.) are provided by this one also. Since many of MUS 1.0's subroutines are used by POLY-SPLIT you must have this PROM or its equivalent available, and the variables that you manually initialize for MUS 1.0 (see LAB NOTES: MUS 1.0, April/ May 1978 Polyphony) must be set for POLY-SPLIT also.

In addition to OUTS, CTRL, etc. which MUS 1.0 used there is a new variable which is unique to POLY-SPLIT; OUT2 (\$BF). This is the variable that tells the program how many channels are to be set aside for use exclusively by AGO keys below the split point. Notice specifically that if MUS 1.0's STG option is selected, the number entered into this variable must include those channels which will be producing envelope transients. (i.e. The number entered for OUT2 will always be an even number when STGs are being used.)

For example, if you have hardware (QuASH, etc.) for eight channels, this number is entered into the normal MUS 1.0 location for it; OUTS (\$EA). If you want to split these into three channels for low keys and five for high keys, you would set OUT2 (\$BF) to contain 03.

The program appears at the end of this column and is loaded starting at location \$000 in the same way that we've loaded programs in the past. If you're the careful sort, you will also save the program on tape as soon as it's loaded so that if there's a problem it won't wipe out all of your work.

When the program has been loaded, preset the MUS 1.0 variables according to your preferences and application, and set the low channels variable (OUT2) as discussed above.

Run the program from location \$000. With POLY-SPLIT running, keys 0 and 1 on the command keyboard retain the functions that they had under MUS 1.0. Key 0 clears and mutes the system; key 1 causes all of the channels to produce a note corresponding to middle C on the AGO keyboard.

A use for command key 2 has now been added; it provides a means of changing the split point while you're playing. Touch this pad and, as long as it's held down, any key on the AGO keyboard that you press will become the new split point. Now while playing, any key below the split point will be assigned to the channels that you've set aside for them, while keys greater or equal to the split point will be assigned to the remaining channels.

```

0010 :*****
0020 :*
0030 :*          POLY-SPLIT          *
0040 :*
0050 :*  A PROGRAM FOR POLYPHONIC    *
0060 :*          SPLIT KEYBOARD      *
0070 :*
0080 :*          BY                    *
0090 :*          JOHN SIMONTON        *
0100 :*
0110 :* (C) 1979 - PAIA ELECTRONICS  *
0120 :*
0130 :*****
0010 KTB1 .DL 00E0
0020 NTBL .DL 00D0
0030 HKEY .DL 00A2
0040 SPLT .DL 00A1
0050 OUT2 .DL 00EC
0060 OUTT .DL 00EB
0070 OUTS .DL 00EA
0080 TRGN .DL 00C3
0090 INIT .DL 0021
0100 NOTE .DL 002B
0110 POLY .DL 0071
0120 DECD .DL 00F0
0130 :
0140 :FIRST, SYSTEM THINGS ARE DISPOSED OF.  THE SYSTEM IS
0150 :INITIALIZED USING MUS 1.0'S "INIT" ROUTINE, THEN THE
0160 :QUASH CHANNELS ARE REFRESHED AND THE AGO KEYBOARD
0170 :SCANNED ALSO USING ROUTINES FROM MUS 1.0
0180 :FINALLY, THE PIEBUG ROUTINE "DECODE" IS USED TO READ THE
0190 :COMMAND KEYBOARD AND ANY COMMANDS ARE EXECUTED.
0200 :0-SYSTEM CLEAR AND RE-INIT; 1-TUNE ALL CHANNELS;
0210 :2-SET SPLIT POINT, ANY AGO KEY PRESSED BECOMES SPLIT
0220 :
0230 :OR 1000
0240 :
1000- A5 EB 0250 STAR LDA *OUTT :GET THE # OF RESERVED LOW CHANS
1002- 85 EC 0260 STA *OUT2 :SAVE PERMANENTLY
1004- A2 07 0270 POSP LDX 07 :SET UP A POINTER/COUNTER
1006- A9 00 0280 SLP9 LDA 00 :AND GET READY TO ZERO STUFF
1008- 95 A2 0290 STA *HKEY,X :ZERO THE TEMPORARY BUFFER
100A- CA 0300 DEX :AND POINT TO THE NEXT
100B- 10 F9 0310 BPL SLP9 :IF SOME ARE LEFT, LOOP
100D- 20 21 00 0320 JSR INIT :MUS 1.0 - INITIALIZE SYSTEM
100E- 20 2B 00 0330 SLP6 JSR NOTE :MUS 1.0 - REFRESH AND READ AGO KBD
1010- 20 00 0F 0340 JSR DECD :PIEBUG - READ COMMAND KEYBOARD
1011- F0 EC 0350 BEQ POSP :IF COMMAND = 0, BRANCH TO RE-INIT
1012- C9 01 0360 CMP 01 :IS COMMAND = 1?
1013- D0 07 0370 BNE NTST :NO, BRANCH TO NEXT TEST
1014- A9 2E 0380 LDA 2E :WILL BECOME MIDDLE C
1015- 20 23 00 0390 JSR INIT+02 :USE PART OF MUS 1.0 INITIALIZE
1016- F0 ED 0400 BEQ SLP6 :BRANCH ALWAYS
1017- C9 02 0410 NTST CMP 02 :IS COMMAND = 2?
1018- D0 08 0420 BNE SLP1 :NO, BRANCH TO POLY-SPLIT PROGRAM
1019- A5 E7 0430 LDA *KTBL+07 :GET THE LOWEST KEY DOWN
1020- F0 E5 0440 BEQ SLP6 :IF NONE ARE DOWN, LOOP
1021- 85 A1 0450 STA *SPLT :SAVE THE KEY AS THE SPLIT POINT
1022- D0 E1 0460 BNE SLP6 :BRANCH ALWAYS

```

```

0470 :
0480 :NOW THE SPLIT PROGRAM. AT THIS POINT A LIST OF THE
0490 :AGO KEYS WHICH THE MUS 1.0 SUBROUTINE "LOOK" FOUND TO
0500 :BE PRESSED HAS BEEN COMPILED AND SAVED IN THE INPUT BUFFER
0510 :AREA "KTBL". WE BEGIN BY REMOVING FROM THE INPUT BUFFER
0520 :ALL THOSE KEYS WHICH ARE ABOVE THE SPLIT POINT AND
0530 :TRANSFERRING THEM TO THE TEMPORARY BUFFER AREA "HKEY".
0540 :
0550 SLP1 LDY 07 :SET UP POINTER TO HIGH BUFFER
0560 LDX 07 :AND ONE TO INPUT BUFFER
0570 SLP0 LDA *KTBL,X :GET THE KEY
0580 BEQ SNX1 :IF ZERO, GO TO NEXT
0590 CMP *SLP1 :GREATER THAN SPLIT POINT?
0600 BCC SNX0 :IF NOT GREATER, BRANCH
0610 STA HKEY,Y :GREATER, SAVE IN HIGH BUFFER
0620 DEY :POINT TO NEXT HIGH KEY BUFFER
0630 LDA 00 :PREPARE AND
0640 STA *KTBL,X :ZERO THIS KEY
0650 SNX0 DEX :POINT TO NEXT KEY
0660 BPL SLP0 :IF SOME LEFT, LOOP
0670 :
0680 :NEXT THE NUMBER OF CHANNELS AVAILABLE FOR LOW KEY USE
0690 :IS TRANSFERRED TO THE TEMPORARY COUNTER "OUTT" AND THE
0700 :MUS 1.0 ALLOCATION PROGRAM POLY IS CALLED TO ASSIGN LOW
0710 :KEYS TO LOW CHANNELS.
0720 :
0730 SNX1 LDA *OUT2 :GET THE NUMBER OF LOW CHANS AVAILABLE
0740 STA *OUTT :AND PUT IT IN THE TEMPORARY COUNTER
0750 JSR POLY+04 :AND CALL THE MAIN PORTION OF POLY
0760 :
0770 :NOW THAT THE LOW KEYS HAVE BEEN ALLOCATED TO LOW CHANNELS,
0780 :THE HIGH KEYS ARE TAKEN FROM "HKEY" AND PLACED BACK IN THE
0790 :INPUT BUFFER (KEYS ALREADY ALLOCATED ARE REMOVED FROM THE
0800 :INPUT BUFFER). SIMULTANEOUSLY THE LOW CHANNELS ARE MOVED
0810 :TO HKEY AND ALL LOW CHANNELS IN THE OUTPUT BUFFER
0820 :ARE MARKED AS "IN USE" SO THAT THEY WILL BE IGNORED
0830 :WHEN HIGH KEYS ARE ALLOCATED.
0840 :
1040- A4 EC 0850 LDY *OUT2 :A COUNTER TO MOVE ONLY THE LOW CHANNELS
104F- A2 07 0860 LDX 07 :AND A POINTER/COUNTER
1051- B5 A2 0870 SLP1 LDA *HKEY,X :GET THE HIGH KEY FROM TEMP BUFFER
1053- 95 E0 0880 STA *KTBL,X :PUT IT IN THE INPUT BUFFER
1055- 88 0890 DEY :ONE LESS LOW CHANNEL TO DO
1056- 30 08 0900 BMI SNX2 :ALL LOW CHANNELS DONE, BRANCH
1058- B5 D8 0910 LDA *NTBL,X :GET THE LOW NOTE
105A- 95 A2 0920 STA *HKEY,X :PUT IT IN TEMPORARY BUFFER
105C- 09 40 0930 ORA 40 :THEN SET THE TRIGGER TO MARK NOTE
105E- 95 D8 0940 STA *NTBL,X :AND REPLACE THE NOTE
1060- CA 0950 SNX2 DEX :ONE LESS CHANNEL, POINT TO NEXT
1061- 10 EE 0960 BPL SLP1 :IF SOME LEFT, LOOP
0970 :
0980 :NOW POLY IS CALLED AGAIN, THIS TIME TO ALLOCATE HIGH CHANNELS
0990 :
1063- 38 1000 SEC :PREPARE FOR SUBTRACTION
1064- A9 10 1010 LDA 10 :16 CHANNELS SUPPORTED BY MUS1
1066- E5 EC 1020 SBC *OUT2 :LESS THE LOW RESERVED CHANNELS
1068- AA 1030 TAX :RESULT IS POINTER
1069- 38 1040 SEC :ANOTHER SUBTRACTION - PREPARE
106A- A5 EA 1050 LDA *OUTS :TOTAL HARDWARE CHANNELS
106C- E5 EC 1060 SBC *OUT2 :LESS LOW RESERVED CHANNELS
106E- 85 EB 1070 STA *OUTT :BECOMES CHANNELS LEFT TO ALLOCATE
1070- 20 77 0D 1080 JSR POLY+06 :CALL MAJOR PORTION OF POLY
1090 :
1100 :FINALLY, THE REAL STATE OF THE LOW CHANNELS IS RESTORED
1110 :TO THE OUTPUT BUFFER. SIMULTANEOUSLY THE TEMPORARY BUFFER
1120 :IS ZERO'D FOR THE NEXT PASS.
1130 :
1073- A4 EC 1140 LDY *OUT2 :NUMBER OF LOW CHANNELS FOR COUNTER
1075- A2 07 1150 LDX 07 :POINTER/COUNTER
1077- 88 1160 SLP2 DEY :ONE LESS LOW CHANNEL
1078- 30 04 1170 BMI SNX3 :AND IF ALL DONE, SKIP NEXT TRANSFER
107A- B5 A2 1180 LDA *HKEY,X :GET THE REAL CHANNEL STATE
107C- 95 D8 1190 STA *NTBL,X :PLACE IN OUTPUT BUFFER
107E- A9 00 1200 SNX3 LDA 00 :NOW GET READY AND
1080- 95 A2 1210 STA *HKEY,X :ZERO THIS TEMPORARY BUFFER LOCATION
1082- CA 1220 DEX :ONE LESS TEMP BUFFER LOCATION
1083- 10 F2 1230 BPL SLP2 :IF SOME REMAIN, LOOP
1085- 30 89 1240 BMI SLP6 :BRANCH ALWAYS TO CONTINUE
1250 :
1260 END .EN

```

```

1280 *****
1290 * NOTES: *
1300 * *
1310 * DUMP PROGRAM FROM 0000-0090 *
1320 * *
1330 * SET THESE LOCATIONS: *
1340 * *
1350 * $0E8 CTRL $40 DYNAMIC *
1360 * $0E9 ODLY $20 DELAY *
1370 * $0EA OUTS $XX TOT CHANS *
1380 * $0EB OUTT $XX LOW CHANS *
1390 * *
1400 * COLD START - $0000 *
1410 * WARM START - $0004 *
1420 * *
1430 * *
1440 * NOTE THE FOLLOWING THINGS: *
1450 * *
1460 * 1) THE PROGRAM IS RELOCATABLE; *
1470 * IT MAY BE LOADED AND RUN IN *
1480 * ANY NON-CONFLICTING MEMORY *
1490 * SPACE *
1500 * *
1510 * 2) CALLING POLY TWICE IS NOT *
1520 * EXTRA EFFICIENT. TIME RE- *
1530 * QUIREMENTS DICTATE MEDIUM *
1540 * TEMPO KNOB SETTING - ABOUT *
1550 * 10 MS/SCAN *
1560 * *
1570 * 3) AS SOON AS THE PROGRAM IS *
1580 * RUNNING, TOUCH COMMAND PAD *
1590 * 2 AND THE KEY WHICH IS TO *
1600 * BE THE SPLIT POINT. THEN 1 *
1610 * TO TUNE AND FINALLY 0 *
1620 * BEFORE PLAYING *
1630 * *
1640 *****
1650 POLY-SPLIT 8.8

```