

DRUMSYS 0.6  
DEVELOPMENTAL VERSION  
USER'S NOTES

Load DRUMSYS 0.6 into the PAIA 8700 Computer/Controller using the following entry sequence:

0-0-0-0-0-0-F-F-0-1-1-1-TAPE

When the program has loaded successfully the display will show "AA". There are at least two copies of the program on the tape. If for any reason the first one will not load, try the second.

When properly loaded, begin the program running from the starting location \$0000 using this sequence:

0-0-0-0-RUN

With the program running, the control keys of the 8700 take on different meanings than that assigned by the PIEBUG Monitor, as outlined below:

DRUM SOUNDS ARE CONTROLLED BY THE KEYS 0-7

<u>KEY#</u>	<u>DRUM SOUND</u>
0	REST (NO DRUM)
1	LIGHT BASS
2	HEAVY BASS
3	SNARE
4	TOM-TOM
5	CONGA
6	WOOD BLOCK
7	CLAVE

Pressing any of the drum sound keys (0-7) causes termination of the current "mode" of operation and reversion to the "DRUM ENTRY" mode. Note that while touching drum sound keys, the corresponding drum sound is produced by the EK-2A and the displays count in hexadecimal. The number shown in the display is the "event number" of the drum sound produced.

In this version of the program, any of the keys 8-F cause the system to be reinitialized. Any score saved in memory when one of these keys is touched will be erased.

The various modes of operation for Drumsys are activated by touching one of the two rows of keys on the 8700 keyboard. Mode names and the corresponding keys that select them are as follows:

<u>KEY NAME</u>	<u>MODE SELECTED</u>
RUN	PLAY
DISP	SET TEMPO
BACK	BACK SPACE
ENTER	STOP/STEP
PCH	CONTINUE
PCL	DUMP SCORE
TAPE	LOAD SCORE
REL	STROBE DRUM

The actions produced by these various modes of operation are as follows:

PLAY - causes the drum score currently in memory to be played at the current tempo rate. Always starts at the beginning of the score (EVENT #0).

SET TEMPO - changes tempo value. When touched, this key causes a counter which will be the tempo value to begin counting. Counting is terminated by touching any other control key. Typically, this control would be used by touching first "TEMPO SET" then "PLAY". The time between touching these two keys is the time between events during playback.

BACK SPACE - causes the program to step through the current score backwards, for editing purposes. In all cases it is important to note that the number shown in the 8700's displays is the event number of the sound just produced.

STOP/STEP - when touched, produces a single step mode of operation. Using the BACK SPACE and STOP/STEP keys allows editing of individual drum sounds. Typical use would be to "STEP" through the score until the drum sound to be replaced (as indicated by sound and event number) is reached. At this point, touching the "BACK" key causes the same drum to sound again (Note that since this is the same "EVENT" as when stepping forward, the displays will not change). The old drum sound may now be replaced with the new simply by touching the proper drum sound key.


CONTINUE - very similar to the "PLAY" key except that the score will pick up from the event currently in the displays.

DRUM SCORE - this command key allows scores in memory to be saved on cassette tapes. When touched, there will be a couple of seconds of apparent inactivity followed by a counting of the displays as the score is transferred from computer memory to tape. Before touching the "DUMP" key, place your recorder in the record mode and allow it to run for several seconds to get beyond the sub-standard tape sections typically found at the beginning of tape cassettes. Note that relays for tape motion are controlled as outlined in the CS-87 manual.

LOAD SCORE - similar to DUMP SCORE except that the memory of the computer is loaded from the cassette tape. Make sure that there is a cassette to be loaded before touching this control as the computer will wait for data transfer completion before continuing with any further action. If this contro is inadvertently touched, you may recover by pressing the reset key and running the program again. There is a soft start location location of sorts at \$014 which can often be used to start the program running without destroying the saved score.

STROBE DRUM - this special effect causes the score to be played at the current tempo rate, but essentially strikes each drum event many times rather than just once. The result is a very unusual bass instrument sounding voice.

**EK-2A / 8700 Wiring Connections**  
For use with Drumsys Program



**NOTES**

```

0010 :
0020 :
0030 :*****
0040 :*
0050 :*          DRUMSYS 0.6
0060 :*
0070 :*      8700/EK-2 DRUM OPERATING
0080 :*          SYSTEM
0090 :*          BY
0100 :*      JOHN SIMONTON
0110 :*
0120 :*(C) 1978 - PAIA ELECTRONICS, INC*
0130 :*
0140 :*****
0150 :
0160 :BUFF .DL 00F0
0170 :CNTR .DL 00EC
0180 :EXP .DL 00EB
0190 :TMPO .DL 00EA
0200 :PNTR .DL 00E9
0210 :DSP .DL 0020
0220 :DECD .DL FF00
0230 :BEEP .DL 0F22
0240 :SCOR .DL 0100
0250 :DUMY .DL 0086
0260 :OUTP .DL 0040
0270 :SNBT .DL 0E25
0280 :CASS .DL 0EAA
0290 :-----
0300 :00E8
0310 :      S-TABLE (CONTROL CODES)
0320 :00E1
0330 :-----
0340 :STBL .DL 00D1
0350 :-----
0360 :00E0
0370 :      DRUM SIGNATURES
0380 :00D9
0390 :-----
0400 :*****
0410 :      .OR 10D1
0420 :*****
0430 :
0440 :TAPE .HS FF00800100010001
0450 :DSIG .HS FFFEFD3F7EFD3FBF
0460 :CRL .HS 6A9D8D877CB2B9CD000820
0470 :
0480 :      .OR 10ED
0490 :PARM .HS FF
0500 :      .OR 10F6
0510 :PAR1 .HS F6F7F8F9FAFBFCFDFF00

```

```

0520 :*****
0530 OR 1000
0540 :*****
0550 :
1000- A9 86 0560 SPHK LDA 86 :SPARE HOOK KEYS 8-F
1002- 85 39 0570 STA *ACTN+01:USED ONLY TO RE-START
1004- EA 0580 NOP :SYSTEM. IN LATER VERSIONS
1005- EA 0590 NOP :WILL PROVIDE ADDITIONAL
0600 : FEATURES
0610 :
1006- A9 00 0620 STAR LDA 0 :PREPARE ACCUMULATOR AND
1008- 85 E9 0630 STA *PNTR :ZERO SCORE POINTER
100A- 8D 20 08 0640 STA DSP :AND DISPLAYS
100D- AA 0650 TAX :PREPARE X REG AS POINTER
100E- 9D 00 01 0660 SLP0 STA SCOR,X :AND USE IT TO CLEAR SCORE
1011- E8 0670 INX
1012- D0 FA 0680 BNE SLP0 :LOOP UNTIL DONE
1014- 20 53 10 0690 SLP1 JSR RDKY :GO READ THE KEYBOARD, ETC.
1017- B0 1F 0700 BCS ACTN :AND IF NO NEW KEYS, BRANCH
1019- C9 10 0710 TSTS CMP 10 :NEW KEY - A "CONTROL" KEY?
101B- B0 16 0720 BCS CTRL :YES - BRANCH TO CONTROL
101D- C9 08 0730 CMP 08 :ONE OF "SPARE" KEYS?
101F- B0 DF 0740 BCS SPHK :YES- BRANCH
0750 : **
1021- A9 86 0760 NTRY LDA 86 :DRUM ENTRY MODE, GET LINK
1023- 85 39 0770 STA *ACTN+01:SET LINK
1025- 89 D9 10 0780 LDA DSIG,Y :GET DRUM SIGNATURE
1028- A6 E9 0790 LDX *PNTR :GET SCORE POINTER
102A- 9D 00 01 0800 STA SCOR,X :SAVE DRUM SIG IN SCORE
102D- 20 3E 10 0810 JSR PLAY :PLAY THE DRUM BEAT
1030- 4C 14 10 0820 JMP SLP1 :LOOP FOR MORE
1033- 89 D1 00 0830 CTRL LDA STBL,Y :GET COMMAND ADDRESS LINK
1036- 85 39 0840 STA *ACTN+01:AND SET LINK IN JSR DUMY
1038- 20 86 00 0850 ACTN JSR DUMY :AND GO TO COMMAND SUBROUTINE
103B- 4C 14 10 0860 JMP SLP1 :THEN LOOP FOR MORE
0870 :
0880 :PLAY SUBROUTINE
0890 :
103E- A4 EB 0900 PLAY LDY *EXP :GET EXPRESSION VARIABLE
1040- 8D 40 08 0910 STA OUTP :OUTPUT CONTROL TO EK-2
1043- 29 7F 0920 AND 7F :RESET STROBE BIT
1045- 88 0930 PLA0 DEY :DELAY FOR THE EXP. TIME
1046- D0 FD 0940 BNE PLA0 :LOOP UNTIL DONE
1048- 8D 40 08 0950 STA OUTP :AND TURN DRUM "OFF"
104B- E6 E9 0960 INC *PNTR :INCREMENT SCORE POINTER
104D- A6 E9 0970 LDX *PNTR :PLACE IN X REGISTER
104F- 8E 20 08 0980 STX DSP :AND SHOW IN DISPLAYS
1052- 60 0990 RTS :THEN RETURN
1000 :
1010 :READ KEY-ALSO IMPORTANT TO TEMPO
1020 :
1053- 20 00 FF 1030 RDKY JSR DECD :PIEBUG KEYBOARD SUBROUTINE
1056- B0 05 1040 BCS DLY :SAME KEY - JUST DELAY
1058- A2 00 1050 LDX 0
105A- 86 EC 1060 STX *CNTR :ZERO TEMPO COUNTER
105C- 60 1070 RTS
105D- A2 20 1080 DLY LDX 20 :SET X AND Y REGISTER
105F- A0 3F 1090 NXTX LDY 3F :DELAY PARAMETERS
1061- 88 1100 DELY DEY :AND DO DELAY.
1062- D0 FD 1110 BNE DELY
1064- CA 1120 DEX

```

```

1065-  D0 F8  1130      BNE NXTX      :LOOP UNTIL DONE
1067-  E6 EC  1140      INC *CNTR      :INCREMENT TEMPO COUNTER
1069-  60      1150      RTS          :AND RETURN
                        1160
                        1170 :RUN SUBROUTINE
                        1180 :
                        1190 :      **
106A-  A9 7C  1200 RUN  LDA 7C      :COMMAND LINK TO "WAIT"
106C-  85 39  1210      STA *ACTN+01 :SET COMMAND LINK
106E-  A9 00  1220 CYCL LDA 00      :PREPARE AND SET
1070-  05 E9  1230      STA *PNTR     :SCORE POINTER TO 0
1072-  A6 E9  1240 CONT LDX *PNTR    :GET CURRENT SCORE POINTER
1074-  BD 00 01 1250      LDA SCOR,X   :GET CURRENT DRUM SIGNATURE
1077-  F0 F5  1260      BEQ CYCL      :ZERO, END OF SCORE-BRANCH
1079-  20 3E 10 1270      JSR PLAY     :GO PLAY DRUM SOUND, ETC.
107C-  A5 EC  1280 WAIT LDA *CNTR    :GET TEMPO COUNTER AND
107E-  45 EA  1290      EOR *TMPO     :COMPARE TO TEMPO VARIABLE
1080-  D0 04  1300      BNE RETN      :IF NOT TIMED OUT, RETURN
1082-  85 EC  1310 0CNT STA *CNTR     :TIMED OUT - ZERO COUNTER
1084-  F0 EC  1320      BEQ CONT      :BRANCH ALWAYS TO PLAY, ETC.
1086-  60      1330 RETN RTS          :RETURN
                        1340
                        1350 :SINGLE STEP SUBROUTINE
                        1360 :
                        1370 :      **
1087-  A9 86  1380 STEP LDA 86      :COMMAND LINK TO "RETN"
1089-  85 39  1390      STA *ACTN+01 :SET COMMAND LINK
108B-  D0 E5  1400      BNE CONT      :BRANCH ALWAYS TO PLAY, ETC.
                        1410
                        1420 :BACKSPACE SUBROUTINE
                        1430 :
                        1440 :      **
108D-  A9 96  1450 BACK LDA 96      :COMMAND LINK TO "NEXT"
108F-  85 39  1460      STA *ACTN+01 :SET COMMAND LINK
1091-  C6 E9  1470      DEC *PNTR     :SCORE POINTER BACK ONE
1093-  D0 DD  1480      BNE CONT      :GO PLAY SCORE, ETC.
1095-  60      1490      RTS          :AND RETURN
                        1500
                        1510 :      **
1096-  A9 86  1510 NEXT LDA 86      :COMMAND LINK TO "RETN"
1098-  85 39  1520      STA *ACTN+01 :SET COMMAND LINK
109A-  C6 E9  1530      DEC *PNTR     :SCORE POINTER BACK ONE
109C-  60      1540      RTS          :RETURN
                        1550
                        1560 :TEMPO
                        1570 :
                        1580 :      **
109D-  A9 A5  1590 TMP  LDA 0A5     :COMMAND LINK TO "NXT2"
109F-  85 39  1600      STA *ACTN+01 :SET COMMAND LINK
10A1-  A9 00  1610      LDA 00      :INITIALIZE TEMPO COUNTER
10A3-  85 EA  1620      STA *TMPO     :AND START COUNTIN
10A5-  E6 EA  1630 NXT2 INC *TMPO    :UNTIL NEXT COMMAND
10A7-  60      1640      RTS          :RETURN
                        1650
                        1660 :SET UP FOR TAPE TRANSFER
                        1670 :
10A8-  A2 07  1680 STTP LDX 07      :TRANSFER SEVEN BYTES
10AA-  B5 D1  1690 STP  LDA *TAPE,X :GET PARAMETER
10AC-  95 F0  1700      STA *BUFF,X  :PLACE PARAMETER
10AE-  CA      1710      DEX          :POINT TO NEXT
10AF-  D0 F9  1720      BNE STP      :LOOP UNTIL ALL TRANSFERED

```

```

10B1-    60      1730      RTS      : THEN RETURN
      1740      :
      1750      : TAPE IN AND OUT ROUTINES
      1760      :
10B2-    20 A8 10 1770      TOUT JSR STTP      : SET UP PARAMETERS
10B5-    A9 D0      1780      LDA 0DD      : SET DUMP "SWITCH"
10B7-    D0 05      1790      BNE D0      : BRANCH ALWAYS
10B9-    20 A8 10 1800      TIN JSR STTP      : SET UP PARAMETERS
10BC-    A9 11      1810      LDA 11      : SET LOAD "SWITCH"
10BE-    20 25 0E 1820      DO JSR SNBT      : TURN ON RELAYS
10C1-    20 AA 0E 1830      JSR CASS      : DO CASSETTE ROUTINE
      1840      : **
10C4-    A9 86      1850      LDA 86      : COMMAND LINK TO "RETN"
10C6-    85 39      1860      STA *ACTN+01 : SET LINK
10C8-    18      1870      CLC      : PREPARE FOR BEEP
10C9-    20 22 0F 1880      JSR BEEP      : TURN OFF RELAYS AND BEEP
10CC-    60      1890      RTS      : AND RETURN
      1900      :
      1910      : STROBE DRUM EFFECT
      1920      :
10CD-    C6 E9      1930      STRB DEC *PNTR : PREPARE TO GET SAME DRUM
10CF-    4C 72 10 1940      JMP CONT      : PLAY DRUM
      1950      :
      1960      :
      1970      :
      1980      :
      1990      END . EN

```

---

## NOTES

---