

POT-SHOT

Monitor Listing

```

0150 0200      ;
0160 0200      ;
0170 0200      ;
0180 0200      ; *****
0190 0200      ; *
0200 0200      ; * POT-SHOT CASSETTE SYSTEM *
0210 0200      ; * WRITTEN BY ROGER WALTON *
0220 0200      ; * COPYRIGHT 1977 BY PAIA *
0230 0200      ; * ELECTRONICS, INC. *
0240 0200      ; * VERSION 1.0 *
0250 0200      ; *
0260 0200      ; *****
0270 0200      ;
0280 0200      PORT = $0900      ;CASSETTE I/O PORT
0290 0200      DISPLY = $0820      ;LED DISPLAY
0300 0200      PIEBUG = $0F52      ;DISP ACC; GOTO PIEBUG
0310 0200      BEEP = $0F22      ;BEEP SUB IN PIEBUG
0320 0200      ;
0330 0200      STATUS = $EF      ;INPUT BIT STATUS
0340 0200      CHKSUM = $EE      ;CHECKSUM
0350 0200      PNTER = $F6      ;16 BIT ADDR POINTER
0360 0200      COMAND = $F0      ;LOAD/DUMP COMMAND
0370 0200      IDENT = $F1      ;FILE IDENTIFIER
0380 0200      ENDADR = $F2      ;END ADDR
0390 0200      BEGADR = $F4      ;BEGINNING ADDR
0400 0200      ;
0410 0200      ; * = $0E00
0420 0E00      ;
0430 0E00 20 25 0E      START JSR SNDBIT      ;START TAPE
0440 0E03 8C 20 08      STY DISPLY      ;CLEAR DISPLAY
0450 0E06 A5 F4      LDA BEGADR      ;MOVE BEGINNING
0460 0E08 85 F6      STA PNTER      ; ADDR TO POINTER
0470 0E0A A5 F5      LDA BEGADR+1
0480 0E0C 85 F7      STA PNTER+1
0490 0E0E A5 F0      LDA COMAND      ;GET COMMAND
0500 0E10 F0 07      BEQ MANUAL      ;BR IF COMMAND= "00"
0510 0E12 20 AA 0E      JSR CASS      ;DUMP OR LOAD BLOCK
0520 0E15 18      CLC
0530 0E16 20 22 0F      JSR BEEP      ;STOP TAPE AND REEF
0540 0E19 4C 52 0F      MANUAL JMP PIEBUG      ;RETURN TO MONITOR
0550 0E1C      ;
0560 0E1C      ;
0570 0E1C      ;
0580 0E1C      ; DELAY SUBROUTINE
0590 0E1C      ; THIS SUB DELAYS FOR ONE HALF CYCLE
0600 0E1C      ; (2000 HZ). Y IS CLEARED, X, A, AND
0610 0E1C      ; CARRY ARE PRESERVED.
0620 0E1C      ;

```

0630	0E1C	8D 00 09	DELAY	STA PORT	;UPDATE TONE
0640	0E1F	A0 0F		LDY #15	
0650	0E21	88	DLY	DEY	
0660	0E22	D0 FD		BNE DLY	;BR UNTIL DELAY FINISHED
0670	0E24	60		RTS	;RETURN
0680	0E25				
0690	0E25				
0700	0E25				
0710	0E25				
0720	0E25			SEND BIT SUBROUTINE	
0730	0E25			THIS SUB SENDS THE CARRY BIT TO THE TAPE.	
0740	0E25			A "1" BIT CONSISTS OF 16 CYCLES OF 2000 HZ	
0750	0E25			AND A "0" BIT CONSISTS OF 8. THIS SUB	
0760	0E25			URNS RELAY 1 ON AND RELAY 2 OFF. Y IS	
0770	0E25			CLEARED, X, A, AND CARRY ARE PRESERVED.	
0780	0E25	48	SNDBIT	PHA	;SAVE A
0790	0E26	8A		TXA	
0800	0E27	48		PHA	;SAVE X
0810	0E28	A2 10		LDX #16	;NO. OF CYCLES FOR A "1"
0820	0E2A	B0 02		BCS CYCLE	;BRANCH IF "1" BIT
0830	0E2C	A2 08		LDX #8	;NO. OF CYCLES FOR A "0"
0840	0E2E	A9 B0	CYCLE	LDA #%10110000	;TAPE ON; OUTPUT HIGH
0850	0E30	20 1C 0E		JSR DELAY	;UPDATE PORT AND DELAY
0860	0E33	A9 80		LDA #%10000000	;TAPE ON; OUTPUT LOW
0870	0E35	20 1C 0E		JSR DELAY	;UPDATE PORT AND DELAY
0880	0E38	CA		DEX	;LAST CYCLE?
0890	0E39	D0 F3		BNE CYCLE	;BRANCH IF NOT
0900	0E3B	A9 90		LDA #%10010000	;TAPE ON; OUTPUT NEUTRAL
0910	0E3D	A2 10		LDX #16	;DELAY COUNTER
0920	0E3F	20 1C 0E	GAP	JSR DELAY	;PRODUCE A GAP
0930	0E42	CA		DEX	;DELAY FINISHED?
0940	0E43	D0 FA		BNE GAP	;BRANCH IF NOT
0950	0E45	68		PLA	
0960	0E46	AA		TAX	;RESTORE X
0970	0E47	68		PLA	;RESTORE A
0980	0E48	60		RTS	;RETURN
0990	0E49				
1000	0E49				
1010	0E49				
1020	0E49				
1030	0E49			DETECT BIT SUBROUTINE	
1040	0E49			THIS SUB WILL PICK UP ONE BIT FROM	
1050	0E49			THE TAPE AND RETURN WITH IT IN THE	
1060	0E49			CARRY FLAG. Y IS Clobbered, X AND A	
1070	0E49			ARE PRESERVED. RELAYS ARE NOT AFFECTED.	
1080	0E49	48	DETBIT	PHA	;SAVE A
1090	0E4A	8A		TXA	
1100	0E4B	48		PHA	;SAVE X
1110	0E4C	AD 00 09	TONE	LDA PORT	;LOOK FOR START OF TONE
1120	0E4F	10 FB		BPL TONE	;BRANCH UNTIL FOUND
1130	0E51	A2 00		LDX #0	;CLEAR COUNTER
1140	0E53	E8	COUNT	INX	;COUNT TRANSITIONS
1150	0E54	A0 23		LDY #35	;TIME LIMIT
1160	0E56	AD 00 09		LDA PORT	;CHECK INPUT
1170	0E59	85 EF		STA STATUS	;SAVE INPUT STATUS
1180	0E5B	AD 00 09	CHECK	LDA PORT	;CHECK INPUT

1190	0E5E	45 EF	EOR STATUS	;	;	;HAS IT SWITCHED?
1200	0E60	30 F1	BMI COUNT	;	;	;IF SO, BRANCH
1210	0E62	88	DEY	;	;	;TIME UP?
1220	0E63	D0 F6	BNE CHECK	;	;	;IF NOT, BRANCH
1230	0E65	E0 08	CPX #8	;	;	;DOES TONE BURST QUALIFY?
1240	0E67	90 E3	BCC TONE	;	;	;BRANCH IF NOT
1250	0E69	E0 18	CPX #24	;	;	;SEC IF "1"; CLC IF "0"
1260	0E6B	68	PLA	;	;	
1270	0E6C	AA	TAX	;	;	;RESTORE X
1280	0E6D	68	PLA	;	;	;RESTORE A
1290	0E6E	60	RTS	;	;	;RETURN
1300	0E6F		;	;	;	
1310	0E6F		;	;	;	
1320	0E6F		;	;	;	
1330	0E6F		;	;	;	
1340	0E6F		;	;	;	SEND BYTE SUBROUTINE
1350	0E6F		;	;	;	THIS SUB SENDS THE BYTE CONTAINED
1360	0E6F		;	;	;	IN THE ACC TO THE TAPE ALONG WITH
1370	0E6F		;	;	;	A START BIT AND ONE STOP BIT. X AND
1380	0E6F		;	;	;	Y ARE CLEARED, A IS PRESERVED,
1390	0E6F		;	;	;	CARRY IS SET. RELAY 1 IS TURNED ON,
1400	0E6F		;	;	;	RELAY 2 IS TURNED OFF.
1410	0E6F	18	SNDBYT CLC	;	;	
1420	0E70	20 25 0E	JSR SNDBIT	;	;	;SEND START BIT
1430	0E73	A2 09	LDX #9	;	;	;SET BIT COUNTER TO 9
1440	0E75	38	SEC	;	;	;SET STOP BIT
1450	0E76	2A	NEXT1 ROL A	;	;	;MOVE BIT TO CARRY
1460	0E77	20 25 0E	JSR SNDBIT	;	;	;SEND IT
1470	0E7A	CA	DEX	;	;	;LAST BIT?
1480	0E7B	D0 F9	BNE NEXT1	;	;	;BRANCH IF NOT
1490	0E7D	60	RTS	;	;	;RETURN
1500	0E7E		;	;	;	
1510	0E7E		;	;	;	
1520	0E7E		;	;	;	
1530	0E7E		;	;	;	GET BYTE SUBROUTINE
1540	0E7E		;	;	;	THIS SUB WILL PICK UP A BYTE FROM
1550	0E7E		;	;	;	TAPE AND RETURN IT IN THE ACC.
1560	0E7E		;	;	;	X AND Y ARE CLEARED, CARRY CONTAINS
1570	0E7E		;	;	;	THE STOP BIT. RELAYS ARE NOT AFFECTED.
1580	0E7E		;	;	;	
1590	0E7E	20 49 0E	GETBYT JSR DETBIT	;	;	;LOOK FOR START BIT
1600	0E81	B0 FB	BCS GETBYT	;	;	;BRANCH UNTIL FOUND
1610	0E83	A2 09	LDX #9	;	;	;SET BIT COUNTER TO 9
1620	0E85	2A	NEXT2 ROL A	;	;	;MOVE BIT TO ACC
1630	0E86	20 49 0E	JSR DETBIT	;	;	;GET NEXT BIT
1640	0E89	CA	DEX	;	;	;LAST BIT?
1650	0E8A	D0 F9	BNE NEXT2	;	;	;BRANCH IF NOT
1660	0E8C	60	RTS	;	;	;RETURN

```

1670 0E8D      ;
1680 0E8D      ;
1690 0E8D      ;
1700 0E8D      ; CHECK ADDRESS SUBROUTINE
1710 0E8D      ; THIS SUB COMPARES THE POINTER TO
1720 0E8D      ; THE END ADDRESS AND SETS THE CARRY
1730 0E8D      ; IF THEY ARE THE SAME. IF THEY ARE
1740 0E8D      ; NOT, THE CARRY IS CLEARED AND POINTER
1750 0E8D      ; IS INCREMENTED. IT ALSO ADDS THE
1760 0E8D      ; CONTENTS OF THE ACC TO THE CHECKSUM
1770 0E8D      ; AND DISPLAYS THE LOW BYTE OF POINTER.
1780 0E8D      ; A IS MUTILATED, X AND Y ARE PRESERVED.
1790 0E8D      ; RELAYS ARE NOT AFFECTED.
1800 0E8D      ;
1810 0E8D D8    CHKADD CLD
1820 0E8E 18    CLC
1830 0E8F 65 EE  ADC CHKSUM      ;UPDATE CHECKSUM
1840 0E91 85 EE  STA CHKSUM      ;SAVE IT
1850 0E93 A5 F6  LDA PNTER       ;GET POINTER LOW
1860 0E95 8D 20 08 STA DISPLY     ;DISPLAY IT
1870 0E98 C5 F2  CMP ENDADR      ;CMP WITH END ADDR LOW
1880 0E9A D0 06  BNE INCPTR      ;BRANCH IF NOT EQUAL
1890 0E9C A5 F7  LDA PNTER+1     ;GET POINTER HIGH
1900 0E9E C5 F3  CMP ENDADR+1    ;CMP WITH END ADDR HIGH
1910 0EA0 F0 07  BEQ RET         ;BRANCH AND SEC IF SAME
1920 0EA2 E6 F6  INCPTR INC PNTER ;INC LOW BYTE
1930 0EA4 D0 02  BNE SKIP2       ;BRANCH IF NO CARRY
1940 0EA6 E6 F7  INC PNTER+1     ;INC HIGH BYTE
1950 0EA8 18    SKIP2 CLC
1960 0EA9 60    RET             ;RETURN
1970 0EAA      ;
1980 0EAA      ;
1990 0EAA      ;
2000 0EAA      ;
2010 0EAA 10 21 CASS BPL LOAD      ;BR IF COMMAND= "LOAD"
2020 0EAC      ;
2030 0EAC      ;
2040 0EAC      ;
2050 0EAC      ; DUMP BLOCK SUBROUTINE
2060 0EAC      ; THIS SUB TRANSFERS A BLOCK OF MEMORY
2070 0EAC      ; TO TAPE. BEGINNING OF THE BLOCK IS
2080 0EAC      ; SPECIFIED WITH "PNTER", END OF THE
2090 0EAC      ; BLOCK WITH "ENDADR", AND BLOCK
2100 0EAC      ; IDENTIFICATION WITH "IDENT". A
2110 0EAC      ; CHECKSUM IS SENT AT THE END OF
2120 0EAC      ; THE BLOCK. A, X, AND Y ARE CLEARD,
2130 0EAC      ; CARRY IS SET. RELAY 1 IS TURNED ON,
2140 0EAC      ; RELAY 2 IS TURNED OFF.
2150 0EAC      ;
2160 0EAC A2 FF  DUMP LDX #255     ;SET UP BIT COUNTER
2170 0EAE 38    NEXT3 SEC         ;"1" BIT
2180 0EAF 20 25 0E JSR SNDBIT    ;SEND LEADER
2190 0EB2 CA    DEX             ;FINISHED?
2200 0EB3 D0 F9  BNE NEXT3      ;BRANCH IF NOT
2210 0EB5      ;
2220 0EB5 86 EE  STX CHKSUM      ;CLEAR CHECKSUM
2230 0EB7 A5 F1  LDA IDENT      ;GET IDENTIFIER

```

```

2240 0EB9 20 6F 0E      JSR SNDBYT      ;SEND IT
2250 0EBC                ;
2260 0EBC A1 F6          NEXT4 LDA (PNTER,X)      ;GET BYTE
2270 0EBE 20 6F 0E      JSR SNDBYT      ;SEND IT
2280 0EC1 20 8D 0E      JSR CHKADD      ;LAST BYTE?
2290 0EC4 90 F6          BCC NEXT4      ;BRANCH IF NOT
2300 0EC6                ;
2310 0EC6 A5 EE          LDA CHKSUM      ;GET CHECKSUM
2320 0EC8 20 6F 0E      JSR SNDBYT      ;SEND IT
2330 0ECB 8A            TXA              ;CLEAR ACC
2340 0ECC 60            RTS              ;RETURN
2350 0ECD                ;
2360 0ECD                ;
2370 0ECD                ;
2380 0ECD                ;
2390 0ECD                ;
2400 0ECD                ;
2410 0ECD                ;
2420 0ECD                ;
2430 0ECD                ;
2440 0ECD                ;
2450 0ECD                ;
2460 0ECD                ;
2470 0ECD                ;
2480 0ECD                ;
2490 0ECD                ;
2500 0ECD                ;
2510 0ECD A2 14          LOAD LDX #20      ;SEARCH FOR LEADER
2520 0ECF 20 49 0E      LOOP JSR DETBIT    ;GET A BIT
2530 0ED2 90 F9          BCC LOAD        ;START OVER IF "0" BIT
2540 0ED4 CA            DEX              ;20 BITS YET?
2550 0ED5 D0 F8          BNE LOOP        ;BRANCH IF NOT
2560 0ED7 20 7E 0E      JSR GETBYT      ;GET IDENTIFIER FROM TAPE
2570 0EDA 8D 20 08      STA DISPLY      ;DISPLAY IT
2580 0EDD A4 F1          LDY IDENT      ;GET DESIRED ID
2590 0EDF F0 04          BEQ SKIP3      ;SKIP COMPARISON IF ID=00
2600 0EE1 C5 F1          CMP IDENT      ;CORRECT ID?
2610 0EE3 D0 E8          BNE LOAD        ;START OVER IF NOT
2620 0EE5 86 EE          SKIP3 STX CHKSUM ;CLEAR CHECKSUM
2630 0EE7                ;
2640 0EE7 20 7E 0E      NEXT5 JSR GETBYT    ;GET A BYTE
2650 0EEA 90 10          BCC ERROR      ;BRANCH IF STOP BIT=0
2660 0EEC 81 F6          STA (PNTER,X)  ;STORE BYTE BY POINTER
2670 0EEE 20 8D 0E      JSR CHKADD      ;LAST BYTE?
2680 0EF1 90 F4          BCC NEXT5      ;IF NOT, GET NEXT BYTE
2690 0EF3                ;
2700 0EF3 20 7E 0E      JSR GETBYT      ;GET CHECKSUM
2710 0EF6 A0 AA          LDY #$AA      ;A-OK MESSAGE
2720 0EF8 C5 EE          CMP CHKSUM     ;IS CHECKSUM OK?
2730 0EFA F0 02          BEQ SKIP4      ;SKIP ERROR MESSAGE IF SO
2740 0EFC A0 EE          ERROR LDY #$EE ;ERROR MESSAGE
2750 0EFE 98          SKIP4 TYA          ;XFER MESSAGE TO ACC
2760 0EFF 60            RTS              ;RETURN
2770 0F00                ;
2780 0F00                ;
2790 0F00                ;
2800 0F00                ;
                .END

```

ERRORS = 0000

SYMBOL TABLE

MANUAL	0E19	CASS	0EAA	CYCLE	0E2E	INCPTR	0EA2
RET	0EA9	SKIP2	0EA8	LOAD	0ECD	SKIP3	0EE5
ERROR	0EFC	SKIP4	0EFE	SNDBIT	0E25	PORT	0900
DISPLY	0820	PIEBUG	0F52	BEEP	0F22	STATUS	00FF
CHKSUM	00EE	PENTER	00F6	COMAND	00F0	IDENT	00F1
ENDADR	00F2	BEGADR	00F4	START	0E00	DELAY	0F1C
DLY	0E21	GAP	0E3F	DETBIT	0E49	TONE	0F4C
COUNT	0E53	CHECK	0E5B	SNDBYT	0E6F	NEXT1	0E76
GETBYT	0E7E	NEXT2	0E85	CHKADD	0E8D	DUMP	0EAC
NEXT3	0EAE	NEXT4	0EBC	LOOP	0ECF	NEXT5	0EE7
END	0F00						